
Introduction to GAMS for economic equilibrium problems, Part A

Tools of Economic Analysis - introducing the “third way”

(1) Traditional analytical theory models   (“pencil and paper” technology)

(2) Econometric estimation and testing.

(3) Computer simulation modeling  - the focus of this course
(A) for use in economic theory: allows us to construct and analyze theories that

are much more complex than those tractable in traditional analytical models
(B) for use in calculating counter-factual outcomes (e.g., due to policy changes)

in empirical models
(C) in both cases, for calculating numerical values for welfare changes and

optimal values for policy variables.

(4) Two ways of formulating economic models
(A) as an constrained optimization problem
(B) as a complementarity problem: square system of equations/inequalities  and

unknowns

1. Formulating an economic equilibrium problem

This first example is a simple supply and demand model of a single market, a partial
equilibrium model.  There are two equations, supply and demand, and two variables,
price and quantity.

Economic equilibrium problems are thus represented as a system of n
equations/inequalities in n unknowns.  Our approach is to formulate the equations and
variables as a complementarity problem.  

This involves (a) associating each equation with a particular variable, called the
complementary variable.  (b) if the variables are restricted to be non-negative (prices and
quantities), then the equations are written as weak inequalities.  

If the equation holds as an equality in equilibrium, then the complementary variable is
generally strictly positive.  

If the equation holds as a strict inequality in equilibrium, the complementary variable is
zero.  



3

Consider first supply of good X with price P.   The supply curve exploits the firm’s
optimization decision, equating price with marginal cost: P = MC. 

MC $ P  with the complementarity condition that X $ 0

Note that the price equation is complementary with a quantity variable.

Suppose that COST = aX + (b/2)X2.  Marginal cost is then given by MC = a + bX.  

a + bX $ P  complementary with  X $ 0.

Optimizing consumer utility for a given income and prices will yield a demand function
of the form X = D(P, I) where I is income. 

X $ D(P, I)  with the complementary condition that P $ 0.

 Note that the quantity equation is complementary with a price variable.

We will suppress income and assume a simple function:  X = c + dP where c > 0, d < 0. 

X $ c + dP  complementary with P $ 0.

2. Coding an economic equilibrium problem in GAMS

First, comment statements can be used at the beginning of the code, preceded with a
*.  Actual code is shown in courier font.

* M0A.GMS  introductory model using MCP
* simple supply and demand model (partial equilibrium)

Now we can begin a series of declaration and assignment statements. 

PARAMETERS
 A     intercept of supply on the P axis (MC at Q = 0)
 B     change in MC in response to Q,  this is dP over dQ
 C     intercept of demand on the Q axis (demand at P = 0)
 D     response of demand to changes in price, dQ over dP
 TAX   a tax rate used later for experiments;

Parameters must be assigned values before the model is solved  These parameter values
are referred to as Case 1 below.

A = 2;
C = 6;
B = 1;
D = -1;
TAX = 0;

Now we declare a list of variables.  They must be restricted to be positive to make any
economic sense, so declaring them as “positive variables” tells GAMS to set lower
bounds of zero on these variables. 

POSITIVE VARIABLES
 P
 X;

Now we similarly declare a list of equations.  We can name them anything we want
provided it is a name not otherwise in use or, of course, a keyword. 

EQUATIONS
 DEMAND
 SUPPLY;

Now we specify our equations.  The format is to give the equation name followed by two
periods (full stops).  Then after a space, the equation is written out in the following way,
with =G= GAMS code for “greater than or equal to”.

SUPPLY..    A + B*X  =G=  P;
 
DEMAND..    X  =G=  C + D*P;

Next we need to declare a model: a set of equations and unknowns. The format is the
keyword model, followed by a model name of your choosing.  Here we use equil. 

Next comes a “/” followed by a list of the equation names: each equation ends with a
period followed by the name of the complementary variable.

MODEL EQUIL /SUPPLY.X, DEMAND.P/;

Finally, we need to tell GAMS to solve the model and what software is needed (GAMS
does many other types of problems, such as optimization)

SOLVE EQUIL USING MCP;

This example uses parameter values which generate an “interior solution”, meaning that
both X and P are strictly positive.

Case 2: It is often the case that a good, or rather a particular way to produce or obtain
a good (e.g., via imports) is too expensive relative to some alternative so that this
production or trade activity is not used in equilibrium.  We sometimes say that this
activity is “slack” in equilibrium.  

A = 7;
SOLVE EQUIL USING MCP;

Case 3: The final possibility is that a good or factor of production may be so plentiful
that it commands a zero price in equilibrium

A = -7;
SOLVE EQUIL USING MCP;



Case 1:  interior solution

Supply (MC):
slope = B

Demand: 
slope = 1/D

A C

Figure 1:  Three outcomes of partial equilibrium example

Case 2:  X is too expensive, not 
produce in equilibrium  (X = 0)

Supply (MC)

Demand

A

Case 3:  excess supply, X is a free 
good in equilibrium  (P = 0)

Supply (MC)
Demand

A

P

X

P

X

P

X

3. Counterfactual (comparative statics) experiments

In the following example, we use a parameter called “TAX”, an ad valorem tax on
the input (marginal cost) to production.  To do so, we declare and specify an additional
equation “SUPPLY2" and model “EQUIL2" (GAMS does not allow us to simply rewrite
the equation or use the old model name once we name a new equation).  Three
parameters are declared and used to extract  information after we solve the model.

PARAMETERS
 CONSPRICE   consumer price
 PRODPRICE   producer price (equal to marginal cost)
 TAXREV      tax revenue (note tax base is producer price);

EQUATIONS
 SUPPLY2;

SUPPLY2..    (A + B*X)*(1+TAX)  =G=  P;

MODEL EQUIL2 /SUPPLY2.X, DEMAND.P/;

Before we solve the model, we need to first return the supply intercept to its original
value: parameter changes are kept as permanent unless “undone” in the code.  Then we
resolve the base case (strictly speaking, this is not necessary).  

A = 2;
TAX = 0;
SOLVE EQUIL2 USING MCP;

Now set the tax at 25% and solve the model.

TAX = 0.25;
SOLVE EQUIL2 USING MCP;

After we solve the model, it is common to extract output using GAMS code.  Note that
with the tax specified on inputs, the price being solved for is the consumer price (price
paid by the consumer) not the producer price (price received by the producer).   Producer
price is the same as marginal cost.



10

GAMS stores three values for variables: its current value, its upper bound (infinity in our
case) and its lower bound (zero in our case).  When the modeler wants the value of a
variable, you have to use the notation NAME.L where the L stands for level.  GAMS does
not automatically print out the values of parameters in the solution, so you have to ask
GAMS to DISPLAY these values.

CONSPRICE = P.L;
PRODPRICE = P.L/(1+TAX);
TAXREV = PRODPRICE*TAX*X.L;

DISPLAY CONSPRICE, PRODPRICE, TAXREV;

4. Reading the output

Again, you need to consult the GAMS users’ manual to see how to run the model, and to
read and interpret output once you run the model.  There is a lot of “stuff” in GAMS
listing (output) files, which are stored as FILENAME.LST after you run the model (LST is
for “listing file”).  Here are just the relevant parts of our model runs

             S O L V E      S U M M A R Y

     MODEL   EQUIL               
     TYPE    MCP                 
     SOLVER  MILES               FROM LINE  50

**** SOLVER STATUS     1 NORMAL COMPLETION         
**** MODEL STATUS      1 OPTIMAL                   

Case 1: The LEVEL is the solution value of the variables.  MARGINAL indicates the
degree to which the equation corresponding to the variable is out of equality.  For P
(price), the equation is DEMAND and the value of the marginal is supply minus demand. 
For X (quantity), the equation is SUPPLY and the value of the marginal is the excess of
marginal cost over price.  Variables that have positive values in the solution should have
zero marginals.  Variables that have zero values in the solution should have positive
marginals.

                       LOWER     LEVEL     UPPER    MARGINAL

---- VAR P               .        4.000     +INF       .     
---- VAR X               .        2.000     +INF       .     

Case 2: This is the zero-output case.  The price equation holds, but the quantity
equation is slack.  The marginal of 1.0 indicates that, at the solution, marginal cost
exceed price by 1.0.

                       LOWER     LEVEL     UPPER    MARGINAL

---- VAR P               .        6.000     +INF       .     
---- VAR X               .         .        +INF      1.000  
   

Case 3: This is the free-good case.  Now the price equation is slack, and the marginal
of 1.0 indicates that, at the solution, supply exceeds demand by 1.0.

                       LOWER     LEVEL     UPPER    MARGINAL

---- VAR P               .         .        +INF      1.000  
---- VAR X               .        7.000     +INF       .     
   



13

Tax example: Here is the tax example (after the supply intercept is set back at 2.0). 
Note the “tax incidence” is split between the producer and consumer: the initial price was
4.0.

                       LOWER     LEVEL     UPPER    MARGINAL

---- VAR P               .        4.444     +INF       .     
---- VAR X               .        1.556     +INF       .     
   
     82 PARAMETER CONSPRICE     =   4.444  consumer price

---- 82 PARAMETER PRODPRICE     =   3.556  producer price 

---- 82 PARAMETER TAXREV        =   1.383  tax revenue

5. The entire model, uninterrupted by talk

* M0.GMS  introductory model using MCP
* simple supply and demand model (partial equilibrium)

PARAMETERS
 A     intercept of supply on the P axis (MC at Q = 0)
 B     change in MC in response to Q,  this is dP over dQ
 C     intercept of demand on the Q axis (demand at P = 0)
 D     response of demand to changes in price, dQ over dP
 TAX   a tax rate used later for experiments;
 
A = 2;
C = 6;
B = 1;
D = -1;

POSITIVE VARIABLES
 P
 X;

EQUATIONS
 DEMAND
 SUPPLY;

SUPPLY..    A + B*X  =G=  P;
 
DEMAND..    X  =G=  C + D*P;

MODEL EQUIL /SUPPLY.X, DEMAND.P/;

SOLVE EQUIL USING MCP;

A = 7;
SOLVE EQUIL USING MCP;

A = -7;
SOLVE EQUIL USING MCP;

PARAMETERS
 CONSPRICE   consumer price
 PRODPRICE   producer price (equal to marginal cost)
 TAXREV      tax revenue (note tax base is producer price);

EQUATIONS
 SUPPLY2;

SUPPLY2..    (A + B*X)*(1+TAX)  =G=  P;

MODEL EQUIL2 /SUPPLY2.X, DEMAND.P/;

A = 2;
TAX = 0;
SOLVE EQUIL2 USING MCP;

TAX = 0.25;
SOLVE EQUIL2 USING MCP;

CONSPRICE = P.L;
PRODPRICE = P.L/(1+TAX);

TAXREV = PRODPRICE*TAX*X.L;
DISPLAY CONSPRICE, PRODPRICE, TAXREV;

Exercise:

(1) Try associating the wrong variables with the equations in the model statement.  Run
the model and see what happens.

(2) Consumer surplus is (geometrically) the area of the triangle under the demand curve
above consumer price.  Define a new parameter CONSURP and assign it the right
value.  Display its value with and without the tax.  

Hints: the intercept of the demand curve on the price axis is -C/D.  

The height of the triangle is (-C/D - P.L), and its base is X.L.  

The area is (-C/D - P.L)*X.L/2.



p

pi+1 = pi - [e'(pi)]-1e(pi)   (generalizes to nxn)

p0

(1)  initial guess:  p0

(2) calculate e(p0) and e'(p0)   (exit if |e(p0)| < ε)

(3) follow gradient path to e = 0

(4) new guess p1 implicitly given by  
      e(p0)/(p0 - p1) = e'(p0)

e(p0)

p1

p2e(p2)

Illustration of gradient method:   find the  zero of the
  excess supply function:   e(p*) = s(p*) - d(p*)  = 0

e(p)

e(p1)

iteration rule:
pi+1 = pi - [e'(pi)]-1e'(pi)

slope = e'(p0)

p*

Introduction to GAMS for economic equilibrium problems, Part B

James R. Markusen
University of Colorado, Boulder

1. Toward General Equilibrium

In the previous section we used a partial-equilibrium model to introduce complementarity
and the GAMS solver.  Now we want to take a step toward general equilibrium,
completing the relationship between producers and consumers.  

Our model has one household and one firm.  The household has a fixed stock of labor
(L).  We make the further assumption that the household derives no utility from leisure,
and so will always supply the entire stock of labor to the market if the wage rate (W) is
positive.  

There is a single good (X) produced from the single input labor.  The firm buys labor
services and sells X to the household at price P.  The household receives income from
selling labor services and uses it all to buy X.

Our model is shown schematically in Figure 2.   As noted in the Figure, general
equilibrium requires five conditions:

(1) the consumer optimizes
(2) the producer optimizes
(3) labor demand equals labor supply (the labor market clears)
(4) X demand equals X supply (the goods market clears)
(5) consumer income equals expenditure (income-expenditure balance)

In our model, the consumer has no alternative use for labor and so optimizes by
supplying it all.  

The consumer prefers more X and so the demand for X is going to be given by X =
WL/P: it is optimal to spend all income on X.

The production function for X will be given as X = "L, where " is the marginal product
of labor in X production.   

We assume competition and free entry, so that the firm is force to price at marginal cost. 
Marginal cost, the cost of producing one more unit of X, is given by W/" (1/" is the
amount of labor needed for one unit of X).



Figure 2:  General equilibrium

Conditions for general equilibrium

(1) Firms optimize

(2) Consumers optimize

(3) Supply equals demand for labor

(4) Supply equals demand for good X

(5) Income equals expenditure (all labor income is spent on X)

Households Firms

Labor supply

Goods supply

Labor income

Expenditure on goods

There are two parameters, total labor supply (LBAR) and productivity " (ALPHA).

PARAMETERS
 LBAR    labor supply (fixed and inelastic)
 ALPHA   productivity parameter  X = ALPHA*L;

LBAR = 100;
ALPHA = 2;

There will be four positive variables: price (P), quantity of X (X), the wage rate (W), and
consumer income (INCOME).

POSITIVE VARIABLES
 P         price of X
 X         quantity of X
 W         wage rate
 INCOME    income from labor supply;

There are four equations.  One important feature of general-equilibrium models with
constant returns to scale as assumed here, is that there is really no “supply” function in
the usual partial-equilibrium sense: at given factor prices, the “supply” curve (the
marginal cost curve) is horizontal.  Supply is perfectly elastic at a fixed price P = MC, the
latter being a constant at constant factor prices.

With constant returns, marginal cost is also average cost, and so this equation can also be
thought of as a free-entry, zero-profits condition.  Accordingly, we refer to this equation
as ZPROFIT (zero profits).

Then we require two market-clearing conditions.  First, supply of X must be greater than
or equal to demand with the complementary variable being P, the price of X.  This is
referred to as CMKTCLEAR (commodity-market clearing).  

Second, the supply of labor must be greater than or equal to its demand, with the
complementary variable being W, the wage rate (price of labor).   This is labeled
LMKTCLEAR (labor-market clearing).  

Finally, we require income balance: consumer expenditure equals labor income.  This is
labeled CONSINCOME.

EQUATIONS
 ZPROFIT      zeroprofits in X production
 CMKTCLEAR    commodity (X) market clearing
 LMKTCLEAR    labor market clearing
 CONSINCOME   consumer income;

Marginal cost is just the wage rate divided by ".  Commodity market clearing requires
that X in equilibrium is greater than or equal to demand, which is just income divided by
P, the price of X.  Labor-market clearing requires that supply, LBAR, is greater than or
equal to labor demand, given by X divided by " (X/").  Income spent on X must equal
wage income, W times LBAR (W*LBAR).

ZPROFIT..    W/ALPHA =G= P;

CMKTCLEAR..  X =G= INCOME/P;

LMKTCLEAR..  LBAR =G= X/ALPHA;

CONSINCOME.. INCOME =G= W*LBAR;

We name the model “GE” for general equilibrium, and associate each equation in the
model with its complementary variable.

MODEL GE /ZPROFIT.X, CMKTCLEAR.P, LMKTCLEAR.W,               
          CONSINCOME.INCOME/;

Now we introduce a new feature, namely, setting starting values of variables.  This helps
the solver find the solution, and can be quite important in complex problems.  The
notation for setting an initial value of a variable is the NAME.L notation we used earlier.

* set some starting values

P.L = 1;
W.L = 1;
X.L = 200;
INCOME.L = 100;

One final issue that is familiar to students of economics, is that there is an indeterminacy
of the “price level” in this type of problem.  If P = W = 1 and INCOME = 100 are

solutions to this model, so are any proportional multiples of these values such as P = W =
2, INCOME = 200.  More formally, there are really only three independent equations in
this model, the fourth is automatically satified if three hold.  This is known in economics
as “Walras’ Law”.  

The solution to this problem is to fix one price, termed the “numeraire”.  Then all prices
are measured relative or in terms of this numeraire.  Suppose that we choose the wage
rate to be the numeraire.  The notation is W.FX where .FX stands for “fix”.  It is
important to understand that W.FX is not the same as W.L.  The former holds W fixed
throughout the remainder of the program, whereas W.L is just setting an initial value that
will be changed by the solver.  When a variable is fixed, GAMS automatically drops the
complementary equation from the model.

* choose a numeraire

W.FX = 1;

Now solve the model.

SOLVE GE USING MCP;

As a counter-factual, double labor productivity and resolve.

* double labor productivity

ALPHA = 4;
SOLVE GE USING MCP;

Here is the solution to the first solve statement.  Notice that W.FX shows up as fixing
both the upper and lower bounds for W.

                       LOWER     LEVEL     UPPER    MARGINAL

---- VAR P               .        0.500     +INF       .     
---- VAR X               .      200.000     +INF       .     
---- VAR W              1.000     1.000     1.000      EPS   
---- VAR INCOME          .      100.000     +INF       .     
   

Here is the solution to the second solve statement.  Doubling of labor productivity allows
twice as much X to be produced from the fixed supply of labor.  With the wage rate fixed
at W = 1, the equilibrium price of X falls in half: twice as much X can be purchased from
the income from a unit of labor.

                       LOWER     LEVEL     UPPER    MARGINAL

---- VAR P               .        0.250     +INF       .     
---- VAR X               .      400.000     +INF       .     
---- VAR W              1.000     1.000     1.000      EPS   
---- VAR INCOME          .      100.000     +INF       .     
   
In order to appreciate correctly the role of the numeraire, suppose instead we wanted to
choose X as numeraire.  The first thing that we have to do is “unfix” W.  This actually
requires two separate statements, since .FX involves fixing both the lower and upper
bounds of W.  Then we set P as numeraire.

W.UP = +INF;
W.LO = 0;
P.FX = 1;

SOLVE GE USING MCP;

* double labor productivity

ALPHA = 4;
SOLVE GE USING MCP;

Here are the solutions for the two runs of our model.

                       LOWER     LEVEL     UPPER    MARGINAL

---- VAR P              1.000     1.000     1.000      EPS   
---- VAR X               .      200.000     +INF       .     
---- VAR W               .        2.000     +INF       .     
---- VAR INCOME          .      200.000     +INF       .     
   

                       LOWER     LEVEL     UPPER    MARGINAL

---- VAR P              1.000     1.000     1.000      EPS   
---- VAR X               .      400.000     +INF       .     
---- VAR W               .        4.000     +INF       .     
---- VAR INCOME          .      400.000     +INF       .     
   

In comparing the two alternative choices of numeraire, note that the choice does not
affect the only “real” variable of welfare significance, the output of X.  The variables
affected are “nominal” variables.  Alternatively, only relative prices matter and the “real
wage”, W/P is the same under either choice of numeraire.

2. A general formulation of general-equilibrium models

The purpose of this section is to discuss a general formulation.

First, general-equilibrium models consist of activities that transform some goods and
factors into others.  These include outputs of goods from inputs, trade activities that
transform domestic into foreign goods and vice versa, activities that transform leisure
into labor supply, and activities that transform goods into utility (welfare).

Activities are most usefully represented by their dual, or cost-functions.  The conditions
for equilibrium are then that marginal cost for each activity is greater than or equal to
price, with the complementary variable being the equilibrium quantity or “level” that
activity.   A quantity variable is complementary to a price equation.   With competition
and constant returns to scale, these are also referred to as zero profit conditions.

Second, general-equilibrium models consist of market clearing conditions.   A
commodity is a general term that includes goods, factor of production, and even utility. 
Thus X and labor are both commodities in our example.  Market clearing conditions
require that the supply of a commodity is greater than or equal to its demand in
equilibrium, with the complementary variable being the price of that commodity.  A price
variable is complementary with a quantity equation.

Finally, there are income-balance equations for each “agent” in a model.  Agents are
generally households, but often include a government sector, or the owner of a firm in
models with imperfect competition.  Expenditure (Exp) equals income for each agent.

Let i subscript activities, also referred to as production sectors.  Let j subscript
commodities, and let k subscript agents or households.  Our general formulation is then:

Inequality Complementary var. No. of eq, unknowns

(1) Zero profits MCi $ Pi Activity (quantity) level i i

(2) Market clearing Sj $ Dj Price of commodity j j

(3) Income balance Expk $ Incomek Income k k

The general-equilibrium system then consists of i+j+k inequalities in i+j+k unknowns: i
activity levels, prices of j commodities, and the incomes/expenditures of k agents.  As per
our earlier discussion, the price level is indeterminate and one price is fixed as a
numeraire.

3. The entire model, uninterrupted by talk

PARAMETERS
 LBAR    labor supply (fixed and inelastic)
 ALPHA   productivity parameter  X = ALPHA*L;

LBAR = 100;
ALPHA = 2;

POSITIVE VARIABLES
 P         price of X
 X         quantity of X
 W         wage rate
 INCOME    income from labor supply;

EQUATIONS
 ZPROFIT      zeroprofits in X production
 CMKTCLEAR    commodity (X) market clearing
 LMKTCLEAR    labor market clearing
 CONSINCOME   consumer income;

ZPROFIT..    W/ALPHA =G= P;

CMKTCLEAR..  X =G= INCOME/P;

LMKTCLEAR..  LBAR =G= X/ALPHA;

CONSINCOME.. INCOME =G= W*LBAR;

MODEL GE /ZPROFIT.X, CMKTCLEAR.P, LMKTCLEAR.W,
CONSINCOME.INCOME/;

* set some starting values

P.L = 1;
W.L = 1;
X.L = 200;
INCOME.L = 100;

* choose a numeraire

W.FX = 1;

OPTION MCP = PATH;
SOLVE GE USING MCP;

* double labor productivity

ALPHA = 4;

SOLVE GE USING MCP;

* change numeraire

W.UP = +INF;
W.LO = 0;
P.FX = 1;

ALPHA = 2;
SOLVE GE USING MCP;

* double labor productivity

ALPHA = 4;
SOLVE GE USING MCP;

Exercise: 

(1) Verify Walras’ law.  Show that if the three equations ZPROFIT, CMKTCLEAR, and
CONSINCOME all hold, then LMKTCLEAR must hold as well.

